DECUS
PROGRAM LIBRARY

DECUS NO.	8-406
TITLE	STATPAC REVISIONS FOR PDP-8/I AND TSS/8
AUTHOR	Dartmouth College, Revisions by Berkshire Community College Submitted by: Roger W. Strickland
COMPANY	Berkshire Community College Pittsfield, Massachusetts
DATE	December 29, 1970
SOURCE LANGUAGE	BASIC

STATPAC REVISIONS FOR PDP-8/I AND TSS/8

DECUS Program Library Write-up DECUS NO. 8-406

ABSTRACT

This package contains 11 programs from the original PDP-10 Dartmouth BASIC Statistical Package which have been revised for the PDP-8/I and TSS/8. The documentation consists of a description and listing of each of the programs.

```
 10 ' BERKSHIRE CØMMUNITY CØLLEGE USER LIBRARY
 20 ' MØDIFIED VERSIØN ØF DARTMØUTH CØLLEGE STATISTICS PRØGRAM
 30 ' DARTMØUTH PRØGRAM NAME: STAT01***
 40 ' MØDIFIED: JULY 2 1970
 90 '
100'  B.C.C. LIBRARY PRØGRAM NAME: STAT01
110'
120'  DESCRIPTIØN--CØMPUTES THE MEAN, VARIANCE, STANDARD DEVIATIØN,
130'  AND THE STANDARD ERRØR ØF THE MEAN FØR ØNE ØR MØRE
140'  SETS ØF DATA.
150'
160'  SØURCE--UNKNØWN
170'
180'  INSTRUCTIØNS--THE DATA START IN LINE NUMBER 900 AND CØNTINUE
190'  NØT PAST LINE 909. THE NUMBER 999999 MUST BE PLACED AT
200'  THE END ØF THE LAST DATUM ØF EACH SET. SAMPLE DATUM ARE IN LINE
210'  900.
220'
230'
240'  *  *  *  *  *  *  *  MAIN PRØGRAM  *  *  *  *  *  *  *  *  *  *
250'
260  READ X
270  IF X = 999999 THEN 320
280 LET S = S + X
290 LET S1 = S1 + X*X
300 LET N = N + 1
310 GØ TØ 260
320 LET M = S/N
330 LET V = (N*S1 -S*S)/N/(N-1)
340 LET S3 = SQR(V)
350 LET S4 = SQR(V/N)
360 PRINT"NUMBER","SUM        SUM-ØF-SQUARES","MEAN","VARIANCE"
370 PRINT N, S, S1, M, V
380 PRINT
390 LET Q = S3/M
400 PRINT "STD. DEV.", "STD. ERRØR ØF THE MEAN","CØEFF. ØF VARIATIØN"
410 PRINT S3, S4, "    ", Q
420 PRINT
430 PRINT
440 READ X
450 IF X = 999999 THEN 500
460 LET S=0
470 LET S1=0
480 LET N=0
490 GØTØ 280
500 STØP
900 DATA 1,2,3,4,5,6,7,8,9,10,999999,4,3,9,15,21,6,999999
910 DATA 999999
920 END

READY
```

```
 10 ' BERKSHIRE CØMMUNITY CØLLEGE USER LIBRARY
 20 ' MØDIFIED VERSIØN ØF DARTMØUTH CØLLEGE STATISTICS PRØGRAM
 30 ' DARTMØUTH PRØGRAM NAME: STATO2***
 40 ' MØDIFIED: JULY 2 1970
 90 '
100 ' B.C.C LIBRARY PRØGRAM NAME: STATO2
110'
120'  DESCRIPTIØN--CØMPUTES THE MEANS, VARIANCES, AND T-RATIØ
130'  FØR TWØ GRØUPS ØF DATA,UNPAIRED. THIS PRØGRAM ASSUMES THAT THE
140'  TWØ GRØUPS HAVE THE SAME VARIANCE.
150'
160'  SØURCE--THØMAS E. KURTZ, KIEWIT CØMPUTATIØN CENTER
170'
180'  INSTRUCTIØNS--PUT DATA IN LINES 900 THRØUGH 998.
190'  END THE FIRST SERIES ØF DATA WITH 999999 AND THEN TYPE
200'  THE SECØND SERIES, AGAIN ENDING WITH 999999.
210'  SAMPLE DATA ARE IN LINES 900 AND 910.
220'
230'
240'
250'  *  *  *  *  *  *   MAIN PRØGRAM  *  *  *  *  *  *  *  *  *  *  *
260 DIM S(2),Z(2),N(2),M(2),V(2),D(2)
270  LET P = 1
280  LET S = 0
290  LET S2 = 0
300  LET N = 0
310  READ X
320  IF X = 999999 THEN 370
330  LET S = S + X
340  LET S2 = S2 + X*X
350  LET N = N + 1
360  GØ TØ 310
370  LET S(P) = S
380  LET Z(P) = S2
390  LET N(P) = N
400  IF P = 2 THEN 430
410  LET P = 2
420  GØ TØ 280
430  REM  NØW WE PRINT THE ANSWERS
440  PRINT "GRØUP", "NUMBER", "MEAN", "VARIANCE", "STD. DEV."
450  PRINT
460  FØR I = 1 TØ 2
470  LET M(I) = S(I)/N(I)
480  LET V(I) = (N(I)*Z(I) - S(I)*S(I))/N(I)/(N(I) - 1)
490  LET D(I) = SQR(V(I))
500  PRINT I, N(I), M(I), V(I), D(I)
510  NEXT I
520  LET Q = V(1)*(N(1)-1)+V(2)*(N(2)-1)
530  LET Q = Q/( N(1) + N(2) -2 ) * ( 1/N(1) + 1/N(2) )
540  LET R = M(1) -M(2)
550  LET W = SQR(Q)
560  PRINT
570  PRINT "MEAN DIFF.", "VAR. DIFF.", "STD. DEV. DIFF."
580  PRINT R,Q,W
590  PRINT
600  PRINT "T RATIØ", R/W, "ØN"; N(1)+N(2)-2; "DEGREES ØF FREEDØM."
610  STØP
900 DATA  160,160,140,190,999999
910 DATA  117,145,147,120,150,120,999999
999 END
READY
```

```
 10 ' BERKSHIRE COMMUNITY COLLEGE USER PROGRAM LIBRARY
 20 ' MODIFIED VERSION OF DARTMOUTH COLLEGE STATISTICS PROGRAM
 30 ' DARTMOUTH PROGRAM NAME: STAT03***
 40 ' MODIFIED: JULY 2 1970
 90 '
100 ' B.C.C. LIBRARY PROGRAM NAME: STAT03
110'
120'  DESCRIPTION--COMPUTES THE MEANS, VARIANCES, AND T-RATIO
130'  FOR TWO GROUPS OF UNPAIRED DATA. THIS PROGRAM ASSUMES THAT THE
140'  GROUPS MAY HAVE UNEQUAL VARIANCES.
150'
160'  SOURCE--THOMAS E. KURTZ, KIEWIT COMPUTATION CENTER
170'
180'  INSTRUCTIONS--PUT DATA IN LINE 900 AND FOLLOWING.
190'  MAKE SURE THE DATA LINE NUMBERS DO NOT EXCEED 998.
200'  END THE FIRST SERIES OF DATA WITH 999999, AND THEN
210'  TYPE IN THE SECOND SERIES,AGAIN ENDING WITH 999999.
220'  SAMPLE DATA ARE IN LINES 900 AND 910.
230'
240'
250'  *  *  *  *  *  *  *   MAIN PROGRAM  *  *  *  *  *  *  *  *  *  **
260 DIM S(2),Z(2),N(2),M(2),V(2),D(2)
270  LET P = 1
280  LET S = 0
290  LET S2 = 0
300  LET N = 0
310  READ X
320  IF X = 999999 THEN 370
330  LET S = S + X
340  LET S2 = S2 + X*X
350  LET N = N + 1
360  GO TO 310
370  LET S(P) = S
380  LET Z(P) = S2
390  LET N(P) = N
400  IF P = 2 THEN 430
410  LET P = 2
420  GO TO 280
430  REM  NOW WE PRINT THE ANSWERS
440  PRINT "GROUP", "NUMBER", "MEAN", "VARIANCE", "STD. DEV."
450  PRINT
460  FOR I = 1 TO 2
470  LET M(I) = S(I)/N(I)
480  LET V(I) = (N(I)*Z(I) - S(I)*S(I))/N(I)/(N(I) - 1)
490  LET D(I) = SQR(V(I))
500  PRINT I, N(I), M(I), V(I), D(I)
510  NEXT I
520  LET Q = V(1) / N(1) + V(2) / N(2)
530  LET W = SQR(Q)
540  LET R = M(1) - M(2)
550  LET K = V(1) / N(1) / Q
560  LET D = 1/( K*K/(N(1) - 1) + (1-K)*(1-K)/(N(2) - 1) )
570  PRINT
580  PRINT "MEAN DIFF.", "VAR. DIFF.", "STD. DEV. DIFF."
590  PRINT R, Q, W
600  PRINT
610  PRINT "T RATIO", R/W ,  "ON"; D;"DEGREES OF FREEDOM."
620  STOP
900  DATA  160, 160, 140, 190, 999999
910  DATA  117, 145, 147, 120, 150, 120, 999999
999 END

READY
```

```
 10 ' BERKSHIRE CØMMUNITY CØLLEGE USER PRØGRAM LIBRARY
 20 ' MØDIFIED VERSIØN ØF DARTMØUTH CØLLEGE STATISTICS PRØGRAM
 30 ' DARTMØUTH PRØGRAM NAME: STAT04***
 40 ' MØDIFIED: JULY 2 1970
 90 '
100 ' B.C.C. LIBRARY PRØGRAM NAME: STAT04
110'
120'  DESCRIPTIØN--CØMPUTES CHI SQUARE STATISTICS FØR
130'  2 BY 2 CØNTINGENCY TABLES.
140'
150'  SØURCE--THØMAS E. KURTZ, KIEWIT CØMPUTATIØN CENTER
160'
170'  INSTRUCTIØNS--DATA IS PUT IN LINE 900 AND FØLLØWING.
180'  ENTER THE TABLES IN NØRMAL ØRDER WITH FØUR ENTRIES ØF
190'  TABLE 1 FIRST, THEN THE 4 ENTRIES FØR TABLE 2, ETC.
200'  DATA LINE NUMBERS SHØULD NØT EXCEED 998. SAMPLE DATA ARE
210'  IN LINE 900.
220'
230'
240'  *  *  *  *  *  *  MAIN PRØGRAM  *  *  *  *  *  *  *  *  *
250'
260   PRINT "TABLE", "      ", "CHI SQUARE"
270   PRINT
280   READ A, B, C, D
290   LET  N = A + B + C + D
300   LET  E = A*D - B*C
310   LET E=ABS(E)
320   LET  F = (E - N / 2 )
330   LET  G = N * F * F
340   LET  R1 = A + B
350   LET  R2 = N - R1
360   LET  C1 = A + C
370  LET  C2 = N - C1
380  LET  X = G / R1 / R2 / C1 / C2
390  PRINT A, B
400  PRINT C, D, X
410  GØ TØ 270
900  DATA  6, 0, 2, 6, 6, 0, 1, 3
999 END

READY
```

```
 10 ' BERKSHIRE CØMMUNITY CØLLEGE USER PRØGRAM LIBRARY
 20 ' MØDIFIED VERSIØN ØF DARTMØUTH CØLLEGE STATISTICS PRØGRAM
 30 ' DARTMØUTH PRØGRAM NAME: STATØ8***
 40 ' MØDIFIED: JULY 2 1970
 90 '
100 ' B.C.C. LIBRARY PRØGRAM NAME: STATØ8
110'
120'  DESCRIPTIØN--CØMPARES TWØ GRØUPS ØF DATA USING THE MEDIAN TEST.
130'
140'  SØURCE--THØMAS E. KURTZ, KIEWIT CØMPUTATIØN CENTER
150'
160'  INSTRUCTIØNS--PUT DATA IN LINE 900 AND FØLLØWING.
170'  THE FIRST DATUM IS THE NUMBER ØF ENTRIES IN THE FIRST
180'  GRØUP, THEN ENTER THE NUMBER ØF ENTRIES IN THE SECØND GRØUP,
190'  THEN THE FIRST GRØUP ITSELF IS ENTERED,AND THEN THE SECØND GRØUP.
200'  THE PRØGRAM PRINTS ØUT THE CHI SQUARE STATISTIC ØF A 2 BY 2 TABLE
210'  ØN 1 DEGREE ØF FREEDØM. DATA SHØULD NØT EXCEED LINE 998.
220'  SAMPLE DATA ARE IN LINE 900.
230'
240'
250'  *  *  *  *  *  *     MAIN PRØGRAM   *  *  *  *  *  *  *  *  *  *
260'
265 DIM A(100)
270    READ M, N
280  LET M1 = M + N
290  LET M2 = INT( M1/2 )
295 FØR Z1= 1 TØ M1
300 READ A(Z1)
305 NEXT Z1
310  LET L = 1
320  LET U = M
330  GØSUB 630
340  LET L = M + 1
350  LET U = M1
360  GØSUB 630
365 LET X = 0
370 LET Y = 0
380  LET I = 1
390  LET J = M + 1
400  FØR K = 1 TØ M2
410     IF A(I) < A(J) THEN 470
420     LET Y = Y + 1
430     LET J = J + 1
440     IF J <= M1 THEN 520
450     LET X = M2 - Y
460     GØ TØ 530
470     LET X = X + 1
480     LET I = I + 1
490     IF I <= M THEN 520
500     LET Y = M2 - X
510     GØ TØ 530
520  NEXT K
530  LET U = M - X
540  LET V = N - Y
550  LET Z = X*V - Y*U
560  LET T = M1*( ABS(Z) - M1/2 ) ↑ 2
570  LET C2 = T / M / N / M2 / (M1 - M2)
580    PRINT "TWØ SAMPLE MEDIAN TEST."
590    PRINT "GRØUP 1  "; X; U
600    PRINT "GRØUP 2  "; Y; V
610    PRINT "CHI-SQUARE = "; C2
620  STØP
630  FØR I = 1 TØ U - L
```

```
 640      LET X = A(L)
 650      LET Q = L
 660      FØR J = L + 1 TØ U - I + 1
 670         IF X >= A(J) THEN 700
 680         LET X = A(J)
 690         LET Q = J
 700      NEXT J
 705 LET J = U-I+1 'THIS STATEMENT PUT IN TSS/8 BECAUSE ØF BUG IN BASIC
 710      IF J = Q THEN 740
 720      LET A(Q) = A(J)
 730      LET A(J) = X
 740   NEXT I
 750   RETURN
 900   DATA  4, 6, 160, 160, 140, 190, 117, 145, 147, 120, 150, 120
 999 END

READY
```

```
 10 ' BERKSHIRE CØMMUNITY CØLLEGE USER PRØGRAM LIBRARY
 20 ' MØDIFIED VERSIØN ØF DARTMØUTH CØLLEGE STATISTICS PRØGRAM
 30 ' DARTMØUTH PRØGRAM NAME: STAT9X***
 40 ' MØDIFIED: JULY 2 1970
 90 '
100 ' B.C.C. LIBRARY PRØGRAM NAME: STAT9X
110'
120'  DESCRIPTIØN--CØMPUTES THE SLØPE AND ØTHER STATISTICS FØR A
130'  LINEAR REGRESSIØN WITH SEVERAL Y VALUES FØR EACH X VALUE.
140'
150'  SØURCE--THØMAS E. KURTZ, KIEWIT CØMPUTATIØN CENTER
160'
170'  INSTRUCTIØNS--PUT DATA STARTING IN LINE 900.
180'  FIRST DATA IS K, THE NUMBER ØF DIFFERENT X-VALUES. THEN
190'  FØR EACH ØF THE K GRØUPS-ENTER THE NUMBER IN THAT GRØUP,
200'  THEN THE CØMMØN X-VALUE, THEN THE Y-VALUES FØR THAT GRØUP.
210'  SAMPLE DATA ARE IN LINES 900-950.
220'
230'
240'  *  *  *  *  *  *  MAIN PRØGRAM  *  *  *  *  *  *  *  *  *
250  READ K
260  FØR J = 1 TØ K
270   READ NO
280   LET N=N+NO
290   READ X
300   FØR I = 1 TØ NO
310    READ Y
320    LET X1 = X1 + X
330   LET Y1 = Y1 + Y
340   LET X2 = X2 + X*X
350   LET Y2 = Y2 + Y*Y
360   LET Z = Z + X*Y
370  NEXT I
380 NEXT J
390 LET S1 = N*X2 - X1*X1
400 LET S2 = N*Z - X1*Y1
410 LET B = S2/S1
420 LET Y3 = Y1/N
430 LET X3 = X1/N
440 LET B1 = Y3 - B*X3
450 LET N1 = N - 1
460 LET N2 = N1 - 1
470 LET S3 = (Y2 - Y1*Y3 - B*S2/N)
480 LET S4 = S3/N2
490 PRINT "NUMBER = "; N, "SLØPE = ";B
500 PRINT "MEAN ØF X = "; X1/N, "  ØF Y = "; Y1/N
510 PRINT "Y-INTERCEPT = "B1
520 PRINT "SUM-ØF-SQUARES", "TØTAL",Y2
530 PRINT "    ", "MEAN  ", Y3*Y1
540 PRINT "    ", "SLØPE ", B*S2/N
550 PRINT "    ", "RESIDUAL ", S3
560 PRINT "STANDARD DEVIATIØNS"
570 PRINT "    ", "X   ", SQR(S1/N/N1)
580 PRINT "    ", "Y   ", SQR((Y2-Y1*Y3)/N1)
590 PRINT "    ", "ERRØR ", SQR(S4)
600 PRINT "    ", "Y-BAR ", SQR(S4/N)
610 PRINT "    ", "SLØPE ", SQR(S4/S1*N)
620 PRINT "    ", "Y-INTER.", SQR(S4*(1/N + X3*X3/S1*N))
630 PRINT "F-RATIØ FØR SLØPE = " B*S2/N/S4
640 STØP
900 DATA 5
910 DATA 3, 4, 6, 8, 7
920 DATA 2, 7, 1, 3
```

```
930 DATA 5, 6, 1, 2, 3, 4, 5
940 DATA 1, 1, 2
950 DATA 4, 5, 2, 4, 6, 8, 10
999 END

READY
```

```
 10 ' BERKSHIRE CØMMUNITY CØLLEGE USER PRØGRAM LIBRARY
 20 ' MØDIFIED VERSIØN ØF DARTMØUTH CØLLEGE STATISTICS PRØGRAM
 30 ' DARTMØUTH PRØGRAM NAME: STAT11***
 40 ' MØDIFIED: JULY 2 1970
 90 '
100 ' B.C.C. LIBRARY PRØGRAM NAME: STAT11
110'
120'  DESCRIPTIØN--CØMPUTES THE SPEARMAN RANK CØRRELATIØN CØEFFICIENT
130'  FØR TWØ SERIES ØF DATA.
140'
150'  SØURCE--THØMAS E. KURTZ, KIEWIT CØMPUTATIØN CENTER
160'
170'  INSTRUCTIØNS--PUT DATA IN LINE 900 AND FØLLØWING.
180'  ENTER DATA IN THE FØLLØWING ØRDER: FIRST THE CØMMØN LENGTH
190'  ØF EACH SERIES, THEN THE ELEMENTS ØF THE SERIES BY GRØUPS
200'  THAT IS X1,Y1,X2,Y2,....ETC.
210'  SAMPLE DATA ARE IN LINE 900.
220'
230'
240'  * * * * * * * *   MAIN PRØGRAM   * * * * * * * * *
250'
260   DIM A(100), B(100)
270   READ N
280  FØR I = 1 TØ N
290      READ A(I), B(I)
300  NEXT I
310  FØR I = 1 TØ N - 1
320     LET X = A(1)
330     LET Y = B(1)
340     LET Q = 1
350     FØR J = 2 TØ N - I + 1
360        IF X >= A(J) THEN 400
370        LET X = A(J)
380        LET Y = B(J)
390        LET Q = J
400     NEXT J
410 LET J = N-I+1
411 IF Q=J THEN 460
420     LET A(Q) = A(J)
430     LET A(J) = X
440     LET B(Q) = B(J)
450     LET B(J) = Y
460  NEXT I
470  REM NØW WE REPLACE THE ELEMENT BY ITS AVERAGE RANK
480  FØR I = 1 TØ N
490     LET X = A(I)
500     FØR J = I + 1 TØ N
510        IF X <> A(J) THEN 540
520     NEXT J
530     LET J = N + 1
540     LET Z = (J + I - 1)/2
550     FØR K = I TØ J - 1
560        LET A(K) = Z
570     NEXT K
580     LET I = J - 1
590  NEXT I
```

```
 600   IF P = 1 THEN 690
 610   REM NØW WE INTERCHANGE A AND B AND SØRT AGAIN.
 620   FØR I = 1 TØ N
 630      LET T = A(I)
 640      LET A(I) = B(I)
 650      LET B(I) = T
 660   NEXT I
 670   LET P = 1
 680   GØ TØ 310
 690   REM NØW WE START CØMPUTING THE CØRRELATIØN CØEFFICIENT.
 700   FØR I = 1 TØ N
 710      LET S1 = S1 + A(I) * A(I)
 720      LET S2 = S2 + B(I) * B(I)
 730      LET S3 = S3 + A(I) * B(I)
 740   NEXT I
 750   LET S = N * N * (N+1) * (N+1) / 4
 760   LET D1 = N * S1 - S
 770   LET D2 = N * S2 - S
 780   LET D = D1 * D2
 790   IF D <> 0 THEN 820
 800   LET R = 0
 810   GØ TØ 830
 820   LET R = (N * S3 - S)/SQR(D)
 830   PRINT "SPEARMAN RANK CØRRELATIØN CØEFFICIENT"
 840   PRINT "  R = ";R
 850   STØP
 900   DATA  5, 480, 56, 500, 61, 520, 78, 540, 71, 56, 82
 910 END

READY
```

```
 10 ' BERKSHIRE CØMMUNITY CØLLEGE USER PRØGRAM LIBRARY
 20 ' MØDIFIED VERSIØN ØF DARTMØUTH CØLLEGE STATISTICS PRØGRAM
 30 ' DARTMØUTH PRØGRAM NAME: STAT20***
 40 ' MØDIFIED: JULY 2 1970
 90 '
100 ' B.C.C. LIBRARY PRØGRAM NAME: STAT20
110 REM JUNE 11, 1970
120 REM  DESCRIPTIØN-- MULTIPLE LINEAR REGRESSIØN ACCØRDING TØ
130 REM  EFRØYSØN'S ALGØRITHM
140 REM
150 REM  SØURCE-- RALSTØN AND WILF,"MATHEMATICAL METHØDS FØR DIGITAL
160 REM  CØMPUTERS",P.191
165 REM  PRØGRAMMED BY THØMAS E. KURTZ, KIEWIT CØMPUTATIØN CENTER
170 REM
180 REM  INSTRUCTIØNS--ENTER DATA STARTING IN LINE 1610.
190 REM  FIRST DATA IS N, THE NUMBER ØF INDEPENDENT VARIABLES, THEN
200 REM  P, THE NUMBER ØF DEPENDENT VARIABLES, THEN M, THE NUMBER
210 REM  ØF DATA SETS, THEN F1, THE VALUE ØF F FØR ENTERING
220 REM  A VARIABLE, AND THEN F2, THE VALUE ØF F FØR REMØVING A
225 REM  VARIABLE. THEN ENTER THE DATA BY SETS, INDEPENDENT VARIABLES
230 REM  BEFØRE DEPENDENT VARIABLES. IF M>50 ØR P>7 ØR N>8 THE
240 REM  DIM STATMENTS IN LINE 300 SHØULD BE CHANGED. SAMPLE DATA
250 REM  ARE IN LINES 1610-1760.
260 REM  BE SURE TØ REMØVE SAMPLE DATA BEFØRE RUNNING PRØGRAM.
270 REM
280 REM  *  *  *  *  *  *  *  MAIN PRØGRAM  *  *  *  *  *  *  *
290 REM
300 DIM S(6,6)
301 DIM A(20,4)
302 DIM  M(8),D(8),B(7),E(7)
310 READ N, P, M, F1, F2
320  LET T1 = 1E-6
330  LET N1 = N + P
340  LET N2 = N1 + N
350  LET K = 1
360  LET D1 = M - 1
370 REM MAT READ A(M,N1)
371 FØR I9 = 1 TØ M
372 FØR I8 = 1 TØ N1
373 READ  A(I9,I8)
375 NEXT I8
376 NEXT I9
380  FØR I = 1 TØ M
390     LET A(I,0)=1
400  NEXT I
410    PRINT "MEANS "
420 FØR I = 0 TØ N1
430     FØR J = I TØ N1
440        LET S = 0
450        FØR L = 1 TØ M
460           LET S = S + A(L,I) * A(L,J)
470        NEXT L
480        LET S(I,J) = S
490     NEXT J
500     LET M(I) = S(0,I) / S(0,0)
510     IF I = 0 THEN 530
520       PRINT M(I),
530  NEXT I
540    PRINT
550    PRINT
560    PRINT "STANDARD DEVIATIØNS"
570  LET M1 = M * D1
580  FØR I = 1 TØ N1
```

```
590      FØR J = I TØ N1
600         LET S(I,J) = ( M*S(I,J) - S(0,I)*S(0,J) )/M1
610      NEXT J
620      LET D(I) = SQR( S(I,I) )
630        PRINT D(I),
640   NEXT I
650     PRINT
660     PRINT
670     PRINT "CØRRELATIØN CØEFFICIENTS"
680   FØR I = 1 TØ N2
690      FØR J = I TØ N2
700         IF J > N1 THEN 740
710         LET S(I,J) = S(I,J) / D(I) / D(J)
720         LET S(J,I) = S(I,J)
730         GØ TØ 800
740         IF I <> J - N1 THEN 780
750         LET S(I,J) = 1
760         LET S(J,I) = -1
770 GØ TØ 800
780         LET S(I,J) = 0
790         LET S(J,I) = 0
800      NEXT J
810   NEXT I
820   FØR I = 1 TØ N1
830      FØR J = 1 TØ N1
840           PRINT S(I,J),
850      NEXT J
860        PRINT
870        PRINT
880   NEXT I
890     PRINT
900   LET K1 = K + N
910     PRINT "DEPENDENT VARIABLE",K
920   FØR I = 1 TØ N
930      LET B(I) = 0
940   NEXT I
950   LET S8 = D(K1) * SQR( S(K1,K1) / D1 )
960 LET V1 = 1E35
970   LET V2 = 0
980   LET N3 = 0
990   LET N4 = 0
1000  FØR I = 1 TØ N
1010     IF ABS( S(I,I) ) <= T1 THEN 1140
1020     LET V0 = S(I,K1) * S(K1,I) / S(I,I)
1030 LET E9= SGN(V0)+2
1031 IF E9 = 1 THEN 1080
1032 IF E9 = 2 THEN 1140
1033 IF E9 = 3 THEN 1040
1040     IF V0 <= V2 THEN 1140
1050     LET V2 = V0
1060     LET N4 = I
1070     GØ TØ 1140
1080   LET I1 = I + N1
1090     LET B(I) = S(I1,K1) * D(K1) / D(I)
1100     LET E(I) = S8 / D(I) * SQR( S(I1,I1) )
1110     IF ABS( V0 ) >= ABS( V1 ) THEN 1140
1120     LET V1 = V0
1130     LET N3 = I
1140  NEXT I
1150  LET S = 0
1160  FØR I = 1 TØ N
1170     LET S = S + B(I) * M(I)
1180  NEXT I
```

```
1190  LET B(0) = M(K1) - S
1200    PRINT "INDEX ", "B  ", "STD. DEV.", "T-RATIO  "
1210    PRINT 0, B(0)
1220  FØR I = 1 TØ N
1230     IF B(I) = 0 THEN 1250
1240       PRINT I, B(I), E(I), B(I) / E(I)
1250  NEXT I
1260    PRINT "STANDARD ERRØR ØF Y  ", S8 * SQR(M-1)
1270    PRINT
1280  LET F = ABS(V1) * D1 / S(K1,K1)
1290  IF F < F2 THEN 1360
1300  LET F = V2 * (D1 - 1) / ( S(K1,K1) - V2 )
1310  IF F <= F1 THEN 1570
1320  LET Q = N4
1330  LET D1 = D1 - 1
1340    PRINT "VARIABLE ENTERING ", Q
1350  GØ TØ 1390
1360  LET Q = N3
1370  LET D1 = D1 + 1
1380    PRINT "VARIABLE LEAVING  ", Q
1390    PRINT "F-LEVEL  ", F
1400  LET Y = 1 / S(Q,Q)
1410  FØR J = 1 TØ N2
1420     LET S(Q,J) = S(Q,J) * Y
1430  NEXT J
1440  LET Y = -Y
1450  FØR I = 1 TØ N2
1460     IF I = Q THEN 1540
1470     LET X = -S(I,Q)
1480     FØR J = 1 TØ N2
1490        IF J = Q THEN 1520
1500        LET S(I,J) = S(I,J) + X * S(Q,J)
1510        GØ TØ 1530
1520        LET S(I,J) = S(I,J) * Y
1530     NEXT J
1540  NEXT I
1550  LET S(Q,Q) = -Y
1560  GØTØ 950
1570  LET K = K + 1
1580  IF K <= P THEN 900
1590    PRINT "   *****  END ØF PRØBLEM  *****"
1600  STØP
1610 DATA  3,1,9,2.5,2.5
1620 DATA 24769,34181,35516,29566
1630 DATA 34181,35516,29566,37009
1640 DATA 35516,29566,37009,34542
1650 DATA 29
566,37009,34542,41485
1660 DATA 37009,34542,41485,49972
1670 DATA 34542,41485,49972,53944
1680 DATA 41485,49972,53944,47062
1690 DATA 49972,53944,47062,47284
1700 DATA 53944,47062,47284,57765
1710 DATA 1, 1, 1, 1
1720 DATA 3,4
1730 DATA 2,2,1,0
1740 DATA 1,3,4
1750 DATA 3,3,0,1
1755 DATA 1,2,3,4
1760 DATA 4,1,0,0
1999 END

READY
```

STAT21

MULTIPLE LINEAR REGRESSION - FANCY OUTPUT

DESCRIPTION - Computes one or more multiple linear regressions on a batch of data.

INSTRUCTION - Place data in line 1940 and following. First data is N (number of data sets or observations), then V (number of variables in data batch), then G (number of regressions to be performed). Then enter the data by the data set, that is enter first the values for all variables at observation 1, then the values for variables at obvervation 2, etc. Variables may be entered in any order, but the order must be the same for each data set. The position of a variable in entering the data is the index of the variables. Thus, if there are four variables entered in the order X, Y, Z, W, the index of Z is 3. Next, for each regression enter (in order) H (the number of the regression), then K (total number of independent variables in the regression), then P1 (enter 1 if you want the residuals printed out, otherwise enter 0), P2 (enter 1 of you want the residuals printed out, otherwise enter 0), and then enter the indexes of the independent variables followed by the index of the dependent variable. If N > 9 or V > 4, then the DIM statements in line 430 and 440 must be changed. Sample data are in lines 1940 through 1992. Be sure to remove the sample data before running the program.

```
 10 ' BERKSHIRE CØMMUNITY CØLLEGE USER PRØGRAM LIBRARY
 20 ' MØDIFIED VERSIØN ØF DARTMØUTH CØLLEGE STATISTICS PRØGRAM
 30 ' DARTMØUTH PRØGRAM NAME: STAT21***
 40 ' MØDIFIED: JULY 2 1970
 90 '
100 ' B.C.C. LIBRARY PRØGRAM NAME: STAT21
110 REM  5/27/70
120 REM
130 REM  DESCRIPTIØN-- CØMPUTES ØNE ØR MØRE MULTIPLE LINEAR
140 REM  REGRESSIØNS ØN A BATCH ØF DATA.
145 REM
150 REM  SØURCE--REVISED 11/16/67 BY PRØF. GERALD CHILDS.
160 REM
170 REM  INSTRUCTIØNS--PLACE DATA IN LINE 1960 AND FØLLØWING.
180 REM  FIRST DATA IS N (NUMBER ØF DATA SETS ØR ØBSERVATIØNS),
190 REM  THEN V (NUMBER ØF VARIABLES IN DATA BATCH), THEN G
200 REM  (NUMBER ØF REGRESSIØNS TØ BE PERFØRMED), THEN ENTER THE
210 REM  DATA BY THE DATA SET, THAT IS ENTER FIRST THE VALUES FØR
220 REM  ALL VARIABLES AT ØBSERVATIØN 1, THEN THE VALUES FØR THE
230 REM  VARIABLES FØR ØBSERVATIØN 2, ETC. VARIABLES MAY BE
240 REM  ENTERED IN ANY ØRDER, BUT THE ØRDER MUST BE THE SAME FØR EACH
250 REM  DATA SET. THE PØSITIØN ØF A VARIABLE IN ENTERING THE DATA IS
260 REM  THE INDEX ØF THE VARIABLE. THUS, IF THERE ARE 4 VARIABLES
270 REM  ENTERED IN ØRDER X,Y,Z,W THE INDEX ØF Z IS 3. NEXT,
280 REM  FØR EACH REGRESSIØN ENTER (IN ØRDER) H (THE NUMBER ØF THE
290 REM  REGRESSIØN), THEN K (TØTAL NUMBER ØF INDEPENDENT VARIABLES
420 REM * * * * * * * * * * MAIN PRØGRAM * * * * * * * * * * * * *
430 DIM X(4,4),A(4,4),D(9,4),Y(4),M(4),S(4)
440 DIM T(4),B(4),U(4,4),R(4,4),C(4,4),Q(9),E(4)
450  READ N, V, G
451 REM    MAT READ D(N,V)
452 FØR I = 1 TØ N
454 FØR J = 1 TØ V
456 READ D(I,J)
458 NEXT J
459 NEXT I
470 FØR I=1 TØ N
480  LET D(I,0)=1
490 NEXT I
500  READ H, K, P1, P2
510 LET M = K + 1
514 REM     MAT READ E(M)
515 FØR Z = 1 TØ M
520 READ E(Z)
525 NEXT Z
530 PRINT "**REGRESSIØN NUMBER"H":DEPENDENT VARIABLE IS"E(M)
540 PRINT
550 IF H>1 THEN 700
560 FØR I=0 TØ V
570  FØR J=0 TØ V
580     LET X=0
590   FØR L=1 TØ N
600       LET X=X+D(L,I)*D(L,J)
610 NEXT L
620    LET X(I,J)=X
630      LET C(I,J)=X
640   NEXT J
650  LET T(I)=X(0,I)/X(0,0)
660 LET B(I)=0
670 IF I=0 THEN 690
680 LET B(I)=SQR(X(I,I)/(N-1)-X(0,I)*X(0,I)/(N*(N-1)))
690 NEXT I
700 PRINT "INDEX", "MEANS", "STANDARD DEVIATIØNS"
```

```
 710 FØR I=1 TØ M
 720 LET M(I)=T(E(I))
 730 LET S(I)=B(E(I))
 740 PRINT E(I), M(I), S(I)
 750 NEXT I
 760 PRINT
 770 PRINT
 780 PRINT "CØRRELATIØN CØEFFICIENTS"
 790  IF H>1 THEN 850
 800 FØR I=1 TØ V
 810  FØR J=1 TØ V
 820   LET R(I,J)=(N*X(I,J)-X(0,I)*X(0,J))/(N*(N-1)*B(I)*B(J))
 830  NEXT J
 840 NEXT I
 850 FØR I=1 TØ M
 860   FØR J=1 TØ M
 870     LET U(I,J)=R(E(I),E(J))
 880   PRINT U(I,J),
 890   NEXT J
 900 PRINT
 910 PRINT
 920 NEXT I
 930 PRINT
 940 LET E(0)=0
 950 FØR I=0 TØ K
 960  LET Y(I)=C(E(I),E(M))
 970   FØR J=0 TØ K
 980    LET X(I,J)=C(E(I),E(J))
 990   NEXT J
1000 NEXT I
1010 FØR I=0 TØ K
1020  FØR J=0 TØ K
1030   IF I<>J THEN 1060
1040   LET A(I,J)=1
1050   GØ TØ 1070
1060   LET A(I,J)=0
1070  NEXT J
1080 NEXT I
1090 FØR I=0 TØ K
1100 IF X(I,I)<1E-6 THEN 1840
1110  LET Y(I)=Y(I)/X(I,I)
1120   FØR J=0 TØ K
1130    LET A(I,J)=A(I,J)/X(I,I)
1140    IF J=I THEN 1160
1150    LET X(I,J)=X(I,J)/X(I,I)
1160   NEXT J
1170  LET X(I,I)=1
1180   FØR L=0 TØ K
1190    IF L=I THEN 1270
1200    LET Y(L)=Y(L)-X(L,I)*Y(I)
1210     FØR J=0 TØ K
1220      LET A(L,J)=A(L,J)-X(L,I)*A(I,J)
1230      IF J=I THEN 1250
1240      LET X(L,J)=X(L,J)-X(L,I)*X(I,J)
1250     NEXT J
1260    LET X(L,I)=0
1270   NEXT L
1280 NEXT I
1290 LET S6=C(E(M),E(M))
```

```
1300 FØR I=0 TØ K
1310   LET S6=S6-Y(I)*C(E(I),E(M))
1320 NEXT I
1330 LET S7=S6/(N-M)
1340 LET R2=1-S7/(S(M)*S(M))
1350 LET R=SQR(R2)
1360 LET S8=SQR(S7)
1370 IF P1=0 THEN 1390
1380 PRINT "VARIANCE-CØVARIANCE MATRIX"
1390 FØR I=0 TØ K
1400   FØR J=0 TØ K
1410  LET A(I,J)=A(I,J)*S7
1420  IF P1=0 THEN 1440
1430 PRINT A(I,J),
1440  NEXT J
1450 IF P1=0 THEN 1480
1460  PRINT
1470  PRINT
1480 NEXT I
1490 PRINT
1500 PRINT "INDEX", "B", "STD. ERRØR", "T-RATIØ"
1510 FØR I=0 TØ K
1520 PRINT E(I), Y(I), SQR(A(I,I)), Y(I)/SQR(A(I,I))
1530 NEXT I
1540 PRINT
1550 PRINT "R-SQUARED=" R2, "R=" R
1560 PRINT
1570 PRINT "STAND. ERRØR ØF EST.=";S8,"D.F.=";(N-M)
1580 PRINT
1590 FØR I=1 TØ N
1600  LET Z=D(I,E(M))-Y(0)
1610   FØR J=1 TØ K
1620   LET Z=Z-Y(J)*D(I,E(J))
1630   NEXT J
1640 LET Q(I)=Z
1650 NEXT I
1660 LET W=0
1670 FØR I=2 TØ N
1680 LET W=W+(Q(I)-Q(I-1))*(Q(I)-Q(I-1))
1690 NEXT I
1700 PRINT
1710 IF P2=0 THEN 1780
1720 PRINT "ACTUAL", "PREDICTED", "RESIDUAL"
1730 LET I = 0
1740 LET I = I + 1
1750  PRINT D(I,E(M)), D(I,E(M))-Q(I), Q(I)
1760 IF I = N THEN 1780
1770 GØ TØ 1740
1780 PRINT
1790 PRINT "DURBIN-WATSØN STAT.=" W/S6
1800 IF H<G THEN 1820
1810 GØ TØ 1850
1820 PRINT
1830  GØ TØ 500
1840 PRINT "CØRRELATIØN MATRIX BECØMING SINGULAR"
1850 PRINT
1860 PRINT "      *****PRØBLEM CØMPLETED*****"
1870 STØP
```

```
1940 DATA 9,4,4
1950 DATA 24769,34181,35516,29566
1952 DATA 34181,35516,29566,37009
1954 DATA 35516,29566,37009,34542
1956 DATA
29566,37009,34542,41485
1958 DATA 37007,34542,41485,49972
1960 DATA 9,4,4
1961 DATA 24769, 34181,35516,29566
1962 DATA 34181, 35516, 29566, 37009
1963 DATA 35516,29566,37009,34542
1964 DATA 29566,37009,34542,41485
1965 DATA 37007,34542,41485,49972
1966 DATA 34542,41485,49972,53944
1967 DATA 41485,49972,53944,47062
1968 DATA 49972, 53944, 47062,47284
1969 DATA 53944,47062,47284,57765
1970 DATA 1, 1, 1, 1
1972 DATA 3,4
1974 DATA 2,2,1,0
1980 DATA 1,3,4
1982 DATA 3,3,0,1
1984 DATA 1,2,3,4
1990 DATA 4,1,0,0
1992 DATA 4,3
2046 END

READY
```

```
  1' BERKSHIRE CØMMUNITY CØLLEGE USER PRØGRAM LIBRARY
  2' MØDIFIED VERSIØN ØF DARTMØUTH STATISTICS PRØGRAM
  3' DARTMØUTH PRØGRAM NAME: R-X-C***
  4' MØDIFIED: JULY 2 1970
  5'
  6' B.C.C. LIBRARY PRØGRAM NAME: RXC
  7'
  8' DESCRIPTIØN-- ANALYSIS ØF TABLE WITH R RØWS AND C CØLUMNS
  9'
 10' SØURCE-- UNKNØWN
 11'
 12' INSTRUCTIØNS-- TYPE RUN FØR INSTRUCTIØNS.
 13'
 14'
 15'
 16'
 70 READ T
 75 DATA 8
 80 DIM Ø(6),H(6)
 85 DIM D(6,6),E(36),F(6,6),R(6),C(6),S(6,6)
 90 DIM Q(6),U(6)
 95 PRINT "HØW MANY RØWS DØES YØUR TABLE HAVE";
100 INPUT A
105 IF A>6 THEN 125
110 PRINT "CØLUMNS";
115 INPUT B
120 IF B<=6 THEN 140
125 PRINT "THIS PRØGRAM WILL NØT ACCØMMØDATE A TABLE LARGER THAN"
130 PRINT "6 X 6.  RE-ENTER NUMBER ØF RØWS"
135 GØ TØ 100
140 PRINT "PLEASE TYPE IN YØUR DATA."
144 PRINT "ENTER TABLE BY RØW, ØNE ENTRY PER LINE  "
145 FØR I = 1 TØ A*B
146 INPUT E(I)
147 NEXT I
150 PRINT
155 PRINT
160 PRINT
165 PRINT
170 PRINT
175 LET N = A*B
180 IF N=(A*B) THEN 200
185 PRINT "YØU HAVE NØT ENTERED THE DATA CØRRECTLY, BUT THAT'S"
190 PRINT "ALL RIGHT, CHIEF, JUST TRY AGAIN."
195 GØ TØ 145
200 PRINT "YØUR ";A;"BY ";B;"TABLE:"
205 PRINT
210 FØR I = 1 TØ A
215 FØR J = 1 TØ B
220 LET R(I) = 0
221 LET C(J) = 0
225 NEXT J
230 NEXT I
235 LET N = 0
```

```
240 FØR I = 1 TØ A
245 FØR J = 1 TØ B
250 LET D(I,J) = E((I-1)*B + J)
255 LET R(I) = R(I) + D(I,J)
260 LET C(J) = C(J) + D(I,J)
265 LET N = N + D(I,J)
270 PRINT D(I,J); TAB(J*T);
275 NEXT J
280 PRINT ":  ";R(I)
285 NEXT I
290 FØR I = 1 TØ ((B*T + 13)/2)
295 PRINT ". ";
300 NEXT I
305 PRINT
310 FØR J = 1 TØ B
315 PRINT C(J); TAB(J*T);
320 NEXT J
325 PRINT ":  ";N
330 PRINT
335 PRINT
340 PRINT "IS THIS DATA TABLE CØRRECT";
345 INPUT A$
350 IF A$ = "NØ" THEN 370
355 IF A$ = "YES" THEN 380
360 PRINT "WHAT";
365 GØ TØ 345
370 PRINT "THEN TYPE THE TABLE IN AGAIN"
375 GØTØ 144
380 FØR I = 1 TØ A
385 FØR J = 1 TØ B
390 LET F(I,J) = R(I)*C(J)/N
395 LET S(I,J) = (D(I,J) - F(I,J))↑2/F(I,J)
400 LET C2 = C2 + S(I,J)
405 NEXT J
410 NEXT I
415 LET G1 = C2
420 LET M = (A-1)*(B-1)
425 LET G = G1/M
430 LET P = 1
435 LET N9 = 1000
440 IF G < 1 THEN 465
445 LET A9 = M
450 LET B9 = N9
455 LET F = G
460 GØ TØ 480
465 LET A9 = N9
470 LET B9 = M
475 LET F = 1/G
480 LET A1 = 2/(9*A9)
485 LET B1 =  2 /(9*B9)
490 LET Z = ABS((1-B1)*F↑(.333333)-1+A1)
495 LET Z = Z/SQR(B1*F↑(.666667)+A1)
500 IF B < 4 THEN 520
505 LET P=(1+Z*(.196854+Z*(.115194+Z*(.000344+Z*.019527))))↑4
510 LET P = .5/P
515 GØ TØ 530
520 LET Z = Z*(1 + .08*Z↑4/B↑3)
525 GØ TØ 505
530 IF G < 1 THEN 540
535 GØ TØ 545
```

```
540 LET P = 1-P
545 LET P1 = INT(10000*P)/10000
550 IF P1 = 0 THEN 565
555 PRINT
560 GØ TØ 590
565 PRINT "THE SIGNIFICANCE LEVEL IS LESS THAN .0001.  THIS IS "
570 PRINT "MUCH RARER THAN HEN'S TEETH IN RANDØM NUMBERS.  "
575 PRINT "IN FACT, YØU MIGHT SAY RARER THAN RUM'S TEETH IN"
580 PRINT "HANDØM RUMBERS!"
585 GØ TØ 715
590 PRINT "THE SIGNIFICANCE LEVEL IS "; P1
595 PRINT
600 LET P = P1
605 IF P > .1 THEN 635
610 IF P > .05 THEN 685
615 IF P > .01 THEN 695
620 IF P > .001 THEN 705
625 PRINT "P < ØR = .001.  RARER THAN HEN'S TEETH IN RANDØM NUMBERS!"
630 GØ TØ 715
635 PRINT "P > .10.   GET MØRE CASES ØR FØRGET IT."
640 PRINT "DØ YØU WANT MØRE ØUTPUT (PERCENTAGE TABLES, GAMMA, ETC.)";
645 INPUT A$
650 IF A$ = "NØ" THEN 1640
655 IF A$ = "YES" THEN 670
660 PRINT "WHAT";
665 GØ TØ 645
670 PRINT "Ø.K.  HERE GØES, BUT I WØULDN'T GIVE YØU TWØ CENTS FØR IT."
675 GØ TØ 715
680 GØ TØ 715
685  PRINT ".10 >= P > .05.  BØRDERLINE.  ØNE IN 11 ISN'T ALL THAT RARE
690 GØ TØ 715
695 PRINT ".05 >= P > .01.  MØST PEØPLE WØULD BUY 20 TØ 1 ØDDS."
700 GØ TØ 715
705 PRINT ".01 .= P > .001.  YØU DØN'T SEE MANY LIKE THAT WITH "
710 PRINT "RANDØM NUMBERS."
715 PRINT
720 PRINT
725 PRINT "WØULD YØU LIKE TØ KNØW THE ABSØLUTE VALUE ØF CHI-SQUARE";
730 INPUT A$
735 IF A$ = "YES" THEN 755
740 IF A$ = "NØ" THEN 775
745 PRINT "CØME ØN, YES ØR NØ";
750 GØ TØ 730
755 PRINT
760 GØSUB 1665
765 PRINT TAB(9);
770 PRINT "CHI-SQUARE IS ";ABS(C2);"WITH ";(A-1)*(B-1);"D.F."
775 PRINT
780 PRINT
785 PRINT "DØ YØU WANT THE PERCENTAGE TABLE";
790 INPUT A$
795 IF A$ = "NØ" THEN 1065
800 PRINT "TYPE 'H' IF YØU WANT IT DØNE HØRIZØNTALLY, 'V' IF"
805 PRINT "VERTICALLY, AND 'B' IF BØTH."
810 INPUT A$
815 IF A$ = "H" THEN 940
820 IF A$ = "V" THEN 845
825 IF A$ = "B" THEN 845
835 PRINT "WHAT";
840 GØ TØ 810
```

```
 845 PRINT "     VERTICAL PERCENTAGE TABLE"
 850 PRINT
 855 FØR I = 1 TØ A
 860 FØR J = 1 TØ B
 865 LET U1 = INT(1000*(D(I,J)/C(J))+.5)/10
 870 LET U(J) = U(J) + U1
 875 PRINT U1;TAB(J*T);
 880 NEXT J
 885 PRINT ":  "; INT(1000*R(I)/N+.5)/10
 890 LET V = V + INT(1000*R(I)/N + .5)/10
 895 NEXT I
 900 FØR I = 1 TØ ((B*T + 13)/2)
 905 PRINT ". ";
 910 NEXT I
 915 PRINT
 920 FØR J = 1 TØ B
 925 PRINT U(J);"%";TAB(J*T);
 930 NEXT J
 935 GØ TØ 945
 940 GØ TØ 960
 945 PRINT ":  ";V;"%"
 950 PRINT
 955 I
F A$ = "V" THEN 1065
 960 PRINT "     HØRIZØNTAL PERCENTAGE TABLE";
 965 PRINT
 970 FØR I = 1 TØ A
 975 FØR J = 1 TØ B
 980 LET T1 = INT(1000*(D(I,J)/R(I))+.5)/10
 985 LET Q(I)=Q(I) + T1
 990 PRINT T1;TAB(J*T);
 995 NEXT J
1000 PRINT ":  ";Q(I);"%"
1005 NEXT I
1010 FØR I = 1 TØ ((B*T + 13)/2)
1015 PRINT ". ";
1020 NEXT I
1025 PRINT
1030 FØR J = 1 TØ B
1035 LET H5=INT(1000*(C(J)/N)+.5)/10
1040 LET W6 = W6 + H5
1045 PRINT H5; TAB(J*T);
1050 NEXT J
1055 PRINT ":  "; W6; "%"
1060 PRINT
1065 PRINT
1070 PRINT
1075 PRINT
1080 PRINT "DØ YØU WANT THE EXPECTED CELL VALUES";
1085 INPUT A$
1090 IF A$ = "NØ" THEN 1220
1095 IF A$ = "YES" THEN 1110
1100 PRINT"WHAT";
1105 GØ TØ 1035
1110 PRINT
1115 PRINT TAB(5);"EXPECTED CELL VALUES"
1120 PRINT
1125 PRINT
1130 FØR I = 1 TØ A
1135 FØR J = 1 TØ B
1140 PRINT INT(100*F(I,J)+.5)/100;TAB(J*T);
```

```
1145 LET H(I) = H(I) + INT(100*F(I,J) +.5)/100
1150 LET Ø(J) = Ø(J) + INT(100*F(I,J) +.5)/100
1155 NEXT J
1160 PRINT ":  "; H(I)
1165 NEXT I
1170 FØR I = 1 TØ ((B*T + 13)/2)
1175 PRINT ". ";
1180 NEXT I
1185 PRINT
1190 FØR J = 1 TØ B
1195 PRINT Ø(J); TAB(J*T);
1200 NEXT J
1205 PRINT ":  "; N
1210 PRINT
1215 PRINT
1220 PRINT
1225 PRINT
1230 PRINT "DØ YØU WANT THE DIFFERENCE BETWEEN THE ØBSERVED VALUE"
1235 PRINT "AND THE EXPECTED VALUE FØR EACH CELL";
1240 INPUT A$
1245 IF A$ = "YES" THEN 1265
1250 IF A$ = "NØ" THEN 1355
1255 PRINT "WHAT";
1260 GØ TØ 1240
1265 PRINT
1270 PRINT
1275 PRINT
1280 PRINT
1285 PRINT "     ØBSERVED VALUE MINUS EXPECTED VALUE"
1290 PRINT
1295 FØR I = 1 TØ A
1300 FØR J = 1 TØ B
1305 PRINT INT(100*(D(I,J)-F(I,J)))/100; TAB(J*T);
1310 NEXT J
1315 PRINT ":  "
1320 NEXT I
1325 FØR I = 1 TØ ((B*T + 7)/2)
1330 PRINT". ";
1335 NEXT I
1340 PRINT
1345 PRINT TAB(B*T);":"
1350 PRINT
1355 PRINT
1360 PRINT
1365 PRINT
1370 PRINT
1375 PRINT "ARE BØTH THE RØW AND CØLUMN SCALES ØRDINAL";
1380 INPUT A$
1385 IF A$ = "NØ" THEN 1650
1390 IF A$ = "YES" THEN 1405
1395 PRINT "WHAT";
1400 GØ TØ 1380
1405 PRINT "DØ YØU WANT TØ KNØW GAMMA";
1410 INPUT A$
1415 IF A$ = "NØ" THEN 1595
1420 IF A$ = "YES" THEN 1440
1425 PRINT "WHAT DID YØU SAY"
1430 PRINT "PLEASE TYPE 'YES' ØR 'NØ'"
1435 GØ TØ 1410
1440 LET G1 = 0
```

```
1445 FØR I = 1 TØ (A - 1)
1450 FØR J = 1 TØ (B - 1)
1455 LET X = 0
1460 FØR K = (I+1) TØ A
1465 FØR L = (J+1) TØ B
1470 LET X = X + D(K,L)
1475 NEXT L
1480 NEXT K
1485 LET G1 = G1 + (D(I,J)*X)
1490 NEXT J
1495 NEXT I
1500 LET G2 = 0
1505 FØR I = 1 TØ (A-1)
1510 FØR J = B TØ 2 STEP -1
1515 LET X = 0
1520 FØR K = (I+1) TØ A
1525 FØR L = (J-1) TØ 1 STEP -1
1530 LET X = X + D(K,L)
1535 NEXT L
1540 NEXT K
1545 LET G2 = G2 + (D(I,J)*X)
1550 NEXT J
1555 NEXT I
1560 LET G9 = (G1-G2) / (G1+G2)
1565 PRINT
1570 GØSUB 1665
1575 PRINT TAB(20);"GAMMA IS "; -INT(10000*G9)/10000
1580 PRINT
1585 PRINT
1590 GØ TØ 1600
1595 PRINT "TSK-TSK.  NØ INTELLECTUAL CURIØSITY."
1600 PRINT "ARE BØTH THE RØWS AND THE CØLUMNS INTERVAL SCALES";
1605 INPUT A$
1610 IF A$ = "NØ" THEN 1660
1615 IF A$ = "YES" THEN 1630
1620 PRINT "HØW'S THAT AGAIN?  A SIMPLE 'YES' ØR 'NØ' WILL SUFFICE."
1625 GØ TØ 1605
1630 PRINT "HAVE YØU CØNSIDERED  PRØDUCT MØMENT CØRRELATIØNS?  I"
1635 PRINT "CAN'T DØ THEM, BUT THERE ARE ØTHER PRØGRAMS WHICH CAN."
1640 PRINT "SØ LØNG.  "
1645 STØP
1650 PRINT "THAT'S ALL I CAN DØ FØR YØU.  BYE-BYE."
1655 STØP
1660 PRINT "THEY AREN'T?  THEN I'LL BE SHØVING ALØNG.  GØØD-BY."
1665 FØR I = 0 TØ 9
1670 PRINT "  ";
1675 NEXT I
1680 REM
1690 REM
1695 RESTØRE
1700 REM
1705 END

READY
```

```
 10 ' BERKSHIRE CØMMUNITY CØLLEGE USER PRØGRAM LIBRARY
 20 ' MØDIFIED VERSIØN ØF DARTMØUTH CØLLEGE STATISTICS PRØGRAM
 30 ' DARTMØUTH PRØGRAM NAME: CØRREL***
 40 ' MØDIFIED: JULY 2 1970
 90 '
100 ' B.C.C. LIBRARY PRØGRAM NAME: CØRREL
110'
120'  DESCRIPTIØN--CØMPUTES A CØRRELATIØN CØEFFICIENT
130'
140'  SØURCE--UNKNØWN
150'
160'  INSTRUCTIØNS--THE DATA CØNSISTS ØF N SETS ØF NUMBERS.
170'  EACH SET CØNSISTS ØF TWØ ELEMENTS, X AND Y.  N IS
180'  ENTERED AS DATA IN LINE 900.  IN LINE 910 AND FØLLØWING
190'  ENTER DATA ALTERNATING X AND Y VALUES.  SAMPLE DATA
200'  ARE IN LINES 900-910.
210'
220'  * * * * * *    MAIN PRØGRAM  * * * * * * * * *
230'
240'
250 READ N
260 FØR I = 1 TØ N
270 READ X, Y
280 LET S1 = S1 + X
290 LET S2 = S2 + Y
300 LET S3 = S3 + X*Y
310 LET S4 = S4 + X*X
320 LET S5 = S5 + Y*Y
330 NEXT I
340 PRINT "THE CØRRELATIØN CØEFFICIENT = ";
350 LET A = N*S3 - S1*S2
360 LET B = SQR((N*S4 - S1↑2)*(N*S5 - S2↑2))
370 LET A = INT(A/B*1000+.5)/1000
380 PRINT A
900 DATA 5
910 DATA 1,5,2,3,3,0,4,-5,5,-11
2046 END

READY
```